




Application Note 25

Title: **How to drive a static LCD from a standard I/O port**
Product Family: **4-bit Microcontroller**

Part Number: EM66xx, EM65xx
Keywords: 4-bit microcontroller, Static LCD, Standard I/O port
Date: March 2, 2005

Several 4 bits uCs do not provide hardware LCD driver module. Therefore it is possible to drive a limited number of segments directly from the standard I/O port, of the micro-controller. The following show how to drive four segments and one Com.

Flow chart:



Data is the variable used to define which segments are active or not



Source Code:

```
; static_LCD.ASM
; The Value of PortA is used to activate each segments driven by PortB
; After each Timer IRQ the Com & segments are inverted
;-----
INCLUDE V6607REG.asm
;-----
; Variables
;-----
STACK0 EQU 030H ; stack variable to save ACCU during interrupt
V_PortC EQU 034H ; temporary variable used for invert Com & segments Value
Data EQU 038H ; temporary variable used to set or reset segments
;-----
; Code Offset
;-----
0000: ORG 0
Reset: JMP Main
;-----
; Interrupt Handler:
;-----
Handler:
H000: STA STACK0 ; save ACCU value (ALSO INDEXES IF USED DURING INTERRUPT)
LDR IRQpC ; clear PC IRQ
LDR IRQpA ; Clear PA IRQ
LDI 04H ; Test Timer IRQ
AND IntRq
JPNZ Timer_IRQ
JMP Hand_End
Timer_IRQ:
LDI 0FH ; invert Com Value
XOR V_PortC
STA V_PortC
XOR Data ; invert the Activate Segments
STA PortB ; Write Segments PB[3:0]
LDI 01H ; Write Com on PC0
AND V_PortC
STA PortC
Hand_End:
LDR STACK0 ; Restore Accu
RTI ; return from Interrupt
;-----
; Main:
;-----
Main:
STI OptReg, 01H ; deactivate WD
STI CIOPB, 0FH ; Set PB in output
STI CPIOB, 02H ; Set PC in output
STI Presc, 08H ; Enable Timer IRQ
STI Beep, 08H ; Enable Timer
STI HTimLS, 01H ; Set the High Value of counter to 01H = 42Hz 0H = 73Hz
STI TimCtr, 09H ; Set 512Hz CLK with autoreload
STI LTimLS, 09H ; Start Count and load 09H = 40Hz 0EH = 73Hz
STI CIRQD, 01H ; Enable General IRQ
Loop: LDR PortA ; Read Data from Port A Save in Data LCD Temp Variable
STA Data
JMP loop
;-----
END
;-----
```